
はじめに 

Koka1（「こうか）と読みます）は、Microsoft 社で開発されている関数型プログラミング

言語です。関数型プログラミング言語は、関数を組み合わせてプログラムを構築するように

設計されたプログラミング言語の総称です。Haskell、F#、Racket などの言語が関数型言

語の特徴をもつ言語として知られています。 

Koka の特徴は「副作用」の扱いにあります（Koka では effect と呼んでおり、この日本語

訳である「効果」が言語名に採用されました）。副作用とは、画面への表示、ファイルの読

み書き、大域的な記憶領域への状態の保存といった、プログラムの実行に影響を与える処理

のことです。例えば、読み込んだファイルの内容によって動作を変えるプログラムを考える

と、ユーザがファイルの内容を書き換えることで、プログラムそのものは変更していないに

もかかわらず、異なる振る舞いをするようになります。一方で、副作用のないプログラムは、

同じ入力に対して同じ結果を常に計算する「書いた通りに動くプログラム」です。Koka は、

副作用の情報を「型」として持つことで、扱いやすくすることを目指して開発されました。 

今回の記事では、Koka を使って簡単なプログラムを書いてみます。記事に掲載するプログ

ラムは、2020年 12月時点で最新の Koka version 2.0.12で動作を確認しています。 

                                                           
 

 

1詳しい情報は以下の公式サイトを参照してください。 

 https://www.microsoft.com/en-

us/research/project/koka/#:~:text=Koka%20is%20a%20function%2Doriented,eve

ry%20function%20is%20automatically%20inferred. 

Koka をインストールする 

Koka は、Linux や MacOS にインストールするのが簡単です。Windows 10 にインストー

ルする場合には、WSL(Windows Subsystem for Linux)2を利用するのがよいでしょう。他

の OS の場合と同様の手順でインストールすることができます。ここでは、筆者が利用して

いるWSL（Ubuntu 18.04）にインストールする手順を解説します。その他の環境でのイン

ストール手順についてはプロジェクトの githubレポジトリ3を参照してください。 

WSL を起動して、公式サイトのインストール手順4に記載されている下記のコマンドを実行

します。なお、Kokaを使うには、GCC5をあらかじめインストールしておく必要があるので

注意してください。GCCのインストール手順は割愛します。 

エラーが発生せずにコマンドが終了すれば、インストールは完了です。 

確認のために、Kokaのインタプリタを起動しましょう。以下のコマンドを実行して、Koka

のロゴが表示されれば成功です。 

初めての Koka プログラミング 

Koka 言語でのプログラミングを実際に試してみましょう。Koka にはコンパイラと対話型

インタプリタが用意されていますが、今回はコンパイラを使います。 

次のプログラムは、文字列 programming language Koka! を画面に出力します。print.kk

という名称のファイルとして保存してください。Koka のソースファイルには.kk という拡

張子をつける決まりです。 

                                                           
 

 

2 https://docs.microsoft.com/ja-jp/windows/wsl/ 

3 https://github.com/koka-lang/koka 

4 https://koka-lang.github.io/koka/doc/book.html 

5 https://gcc.gnu.org/ 

fun main() { 
    println("programing language Koka!") 
} 

> curl -sSL https://github.com/koka-lang/koka/releases/latest/download/install.sh | sh 

> koka 

Kokaでは、様々な計算処理を関数で記述します。print.kkは関数mainのみで構成されてい

ます。関数の定義の先頭には予約語funを付け、その後に関数名mainを書きます。その後

のかっこ()内には関数の引数を書きますが、main関数には引数がないので空欄になってい

ます。関数の範囲は中かっこ{}で定めます。main関数は、C言語のmain関数やJavaのmain

メソッドと同様に、プログラム実行後最初に実行される関数と定められており、プログラ

マは必ず一つ書かなければなりません。main関数内のprintln関数は、引数に指定した文

字列を画面に出力するライブラリ関数です。ここでは、文字列 programing language 

Koka! を画面に出力します。 

次に、koka コマンドでソースファイルをコンパイルします。koka コマンドには-c オプシ

ョンを付けます。 

コンパイルに成功すると、コンパイラの出力の最終行にあるように、実行ファイル

（out/v2.0.12/gcc-debug/print）が生成されます。このファイルを実行すると、作成し

たプログラムを実行することができます。 

コンパイルと実行を一度に行うには、-cコマンドをつけずに kokaコマンドを実行すればよ

いようです。ファイル名を指定するのを忘れないでください。 

> out/v2.0.12/gcc-debug/print 
programing language Koka! 

> koka print.kk 
compile: print.kk 
loading: std/core 
loading: std/core/types 
loading: std/core/hnd 
check  : print 
linking: print 
created: out/v2.0.12/gcc-debug/print 
  
programing language Koka! 

> koka -c print.kk 
compile: print.kk 
loading: std/core 
loading: std/core/types 
loading: std/core/hnd 
check  : print 
linking: print 
created: out/v2.0.12/gcc-debug/print 

クイックソート 

もう少し複雑な例として、整列アルゴリズムの一つであるクイックソートを実装します。ク

イックソートは、値の小さい順、あるいは大きい順に配列やリストを並び替えるアルゴリズ

ムです。ピボットと呼ばれる値より小さい値を集めた部分リストと、大きい値を集めた部分

リストを作り、これら二つの部分リストに対して再帰的にクイックソートを実行することで、

リスト全体を並び替えます。 

整数から成るリスト [3, 2, 4, 6, 1, 5] を値の小さい順に、また、文字列 software の

各文字をアルファベット順に並び替えるプログラムを以下に示します。 

 

このプログラムを quick-sort.kk というファイル名で保存した後、コンパイルして実行して

ください。以下のように、整数リストも文字列も正しく並び替えられたことが分かります。 

 

fun main() { 

    [3, 2, 4, 6, 1, 5].quick-sort((<)).show.println // 整数リストの並び替え 

    "software".list.quick-sort((<)).string.println // 文字列の並び替え 

} 
 

// クイックソート 

fun quick-sort(xs, lt) { 
    match (xs) { 

        Nil -> Nil // 空の場合 

        Cons(piv, xss) -> { // 空でない場合、再帰的に並び替える 

            var less := xss.filter(fn(x) {x.lt(piv)}).quick-sort(lt) 
            var greater := xss.filter(fn(x) {!x.lt(piv)}).quick-sort(lt) 
            less + [piv] + greater 
        } 
    } 
} 

> koka quick_sort.kk 
compile: quick-sort.kk 
loading: std/core 
loading: std/core/types 
loading: std/core/hnd 
check  : quick-sort 
linking: quick_dash_sort 
created: out/v2.0.12/gcc-debug/quick_dash_sort 
  
[1,2,3,4,5,6] 
aeforstw 

関数 quick-sort から見ていきます。この関数には引数が二つあります。xs は並び替えた

いリストで、lt は並び替えの順序を決める値同士の比較関数（小さい順に並べる場合は、ど

ちらの値が小さいかを決定する関数）です。このように、Koka は関数を引数として取るこ

とができます。 

予約語 match はパターンマッチングを表し、xs の値のパターンに応じて処理を切り替えま

す。Kokaにおけるリストは、Nilまたは Cons(x, xss)のどちらかの値になります。Nilは

空リストを意味します。空リストに対しては、並べ替え後も空リストなので、quick-sort

関数は Nil を返します。Kokaは関数の最後に書いた式の値を返します（return Nil とし

ても構いません）。Cons(x, xss)は、空リストではなく、先頭がｘでそれ以降が xssであ

るようなリストを意味します。例えば、リスト [3, 2, 4, 6, 1, 5]の場合には、xが 3、

xssが[2, 4, 6, 1, 5]です。 

リストが空リストでない場合には並び替え処理を実行します。予約語 varは局所変数の定義

です。変数 less にはリストの先頭の値より小さい値（小さい順に並び替える場合）をリス

トとして格納します。filter 関数は Koka のライブラリ関数で、与えられた条件を満たす

値をリスト xs から取り出します。条件は引数として関数 fn(x) {x.lt(piv)}で指定して

います。fn は無名関数という名前を付けない関数の定義に使う予約語です。ここでは、ピ

ボット piv の値より小さい値のみを選んでいます。また、Koka の特徴的な記法であるドッ

ト記法を併せて使っています。filter関数の引数はリスト xsと条件を表す関数 fのため、

filter(xs, f)と書きます。しかし、Koka は第 1 引数 xs を関数名の前に移動した

xs.filter(f)という記法を許しています。他の関数についても同様にドットを使った記法

を使うことができます。この記法により、パイプ処理やオブジェクト指向言語に近い書き方

ができるようになっています。このプログラムの場合、 xss.filter(fn(x) 

{x.lt(piv)}).quick-sort(lt)は quick-sort(filter(xss, fn(x) {x.lt(piv)}), lt)

と同じ意味となります。つまり、リストの先頭の値 pivより小さい値のみを取り出したリス

トを quick-sort関数で再帰的に並び替えます。 

一方、変数 greater には、xsの先頭の値以上の値を格納したリストを並び替えた結果が格

納されます。感嘆符 ! は否定(NOT)演算に相当する関数です。 

最後に、less + [piv] + greaterという式で、正しい順序で並び替えたリストを生成し

て返します。＋はリストの連結をするライブラリ関数です。 

main 関数では、整数リストと文字列を並び替えて画面に表示する処理を行います。ここで

もドット記法を使っている点に注意してください。show はリストを文字列に変換するライ

ブラリ関数、listは文字列を文字のリストに変換するライブラリ関数、stringは文字のリ

ストを文字列に変換するライブラリ関数です。整数リスト、文字列のどちらの場合も、「リ

ストを並び替えた後、文字列に変換してから画面に表示する」という流れで処理が行われま

す。 

副作用を扱うプログラム 

副作用を扱うプログラムも簡単に見ていきます。クイックソートの途中経過を履歴として、

実行順とその逆順に表示するプログラムを以下に示します(quick-sort-out.kkとします)。 

fun main() { 
    print-quick-sort() 
    print-reverse-quick-sort() 
    () 
} 
 

// プログラマが定義する副作用 

effect fun history(xs:some<a>(list<a>)): () 
 

// クイックソート 

fun quick-sort(xs, lt) { 
    match (xs) { 
        Nil -> Nil 
        Cons(piv, xss) -> { 
            var less := xss.filter(fn(x) {x.lt(piv)}).quick-sort(lt) 
            var greater := xss.filter(fn(x) {!x.lt(piv)}).quick-sort(lt) 
            var sorted := less + [piv] + greater 
            history(sorted) 
            sorted 
        } 
    } 
} 
 

// 並び替えの履歴を表示する 

fun print-quick-sort() { 

    // 並び替えの途中経過を表示する 

    with fun history(xss: list<int>) { 
        println(xss.show) 
    } 
    [3, 2, 4, 6, 1, 5].quick-sort((<)) 
} 
 

// 並び替えの履歴を逆順に表示する 

fun print-reverse-quick-sort() { 

    // 並び替えの途中経過を逆順に保存する 

    var reverse := Nil 
    with fun history(xss: list<int>) { 
        reverse := Cons(xss, reverse) 
    } 
    println("Print Reversed History.") 
    [3, 2, 4, 6, 1, 5].quick-sort((<)) 

    // reverseの各値の表示 

    foreach(reverse) fn(e) {println(e.show)} 
} 

実行結果は以下の通りです。 

 

関数 quick-sort には、並び替えた後のリスト sorted を引数とする history(sorted)を

追加しました。これを Effect Handler といい、プログラマが定義した副作用を伴う処理を

実行する箇所です。Effect Handler は quick-sort 関数の上に以下のように定義していま

す。 

副作用の具体的な内容は、クイックソートを呼び出す関数で定義します。履歴を順に表示す

る print-quick-sort関数では、次のように with文を使って、historyを表示処理に束縛

しています。 

Effect Handlerによって、print-quick-sort関数で、画面表示という副作用をプログラマ

が指定できるようになりました。 

> koka quick_sort-out.kk 
compile: quick-sort-out.kk 
loading: std/core 
loading: std/core/types 
loading: std/core/hnd 
check  : quick-sort-out 
linking: quick_dash_sort_dash_out 
created: out/v2.0.12/gcc-debug/quick_dash_sort_dash_out 
  
[1] 
[1,2] 
[5] 
[5,6] 
[4,5,6] 
[1,2,3,4,5,6] 
Print Reversed History. 
[1,2,3,4,5,6] 
[4,5,6] 
[5,6] 
[5] 
[1,2] 
[1] 

// プログラマが定義する副作用 

effect fun history(xs:some<a>(list<a>)): () 

    // 並び替えの途中経過を表示する 

    with fun history(xss: list<int>) { 
        println(xss.show) 
    } 

同様に、逆順に表示する print-reverse-quick-sort 関数では、履歴をリストの先頭に追

加保存する処理を Effect Handler に定義しています。この結果得られる履歴の逆順リスト

を reverseとしました。 

print-reverse-quick-sort 関数のリスト reverse の値を表示する処理は、ライブラリ関

数 foreachを使って書いています。 

foreach 関数は、Python の for 文に似た機能を提供し、与えられたリストの各要素に対し

て繰り返し処理を実行します。foreach 関数の引数は、リスト reverse と、そのリストの

値を引数で受け取って表示する無名関数 fn(e) {println(e.show)} です。ただし、この

例では、第 2引数である無名関数は引数を書くかっこの外に書いています。この記法は引数

書かれた無名関数を見やすくする Kokaの機能で、Trailing Lambdaと呼びます。 

おわりに 

今回は、Microsoft 社が開発しているプログラミング言語 Koka の概要を紹介しました。残

念ながら、副作用を型として扱い、強い型付けを行うなど、Koka の重要な特徴は触れるこ

とができませんでした。この機能の解説は別の機会に譲りたいと思います。 

公式サイトには他にもプログラムの例やチュートリアルが掲載されています。興味のある読

者は試してみてください。 

    // reverseの各値の表示 

    foreach(reverse) fn(e) {println(e.show)} 

